ME 201/MTH 281/ME400/CHE400
Far Field for the Laplace Equation

1. Introduction

In this notebook we consider the far-field behavior of the solution of the boundary value problem given below for the Laplace equation in a two-dimensional semi-infinite region. By far-field behavior, we mean the behavior of the solution as we move far from the boundary on which the inhomogenous boundary condition is specified.

$$\frac{\partial^2 \Phi}{\partial x^2} + \frac{\partial^2 \Phi}{\partial y^2} = 0\ ,\quad 0 < x < L,\ y > 0\ ,$$
$$\text{with } \Phi(0, y) = 0\ ,\ \Phi(L, y) = 0\ ,\ \Phi(x, 0) = f(x),\ \text{and } \Phi \underset{y\to\infty}{\longrightarrow} 0\ . \tag{1}$$

We obtained the solution to this problem in class by separation of variables. The result is

$$\Phi(x, y) = \sum_{n=1}^{\infty} C_n\, e^{-n\pi y/L} \sin(n\pi x/L)\ ,\quad \text{where } C_n = \frac{2}{L}\int_0^L f(x)\sin(n\pi x/L)\, dx\ . \tag{2}$$

For interpretative purposes, it is helpful to compare the series for the solution with the series for the boundary function *f*:

$$f(x) = \sum_{n=1}^{\infty} C_n \sin(n\pi x/L)\ . \tag{3}$$

We may describe the solution (2) in the following terms. The boundary function $f(x)$ is resolved into Fourier modes $\sin(n\pi x/L)$ with amplitude C_n. The solution for Φ is constructed by letting each mode decay with height by the factor $e^{-n\pi y/L}$, and then reassembling those decaying modes into the series (2). We may abstract a simple scaling rule from this result. We observe that for the mode $\sin(n\pi x/L)$, the half wavelength -- which is a reasonable measure of the scale of variation of this function -- is L/n. From the y-dependent part of the solution for Φ, we see that this mode decays with height with an e-folding scale of $L/n\pi$, which is $\frac{1}{\pi}$ times the half-wavelength. Thus high-frequency wiggles on the boundary decay with height much more rapidly than low frequency wiggles. This is the basis for the far-field approximation discussed below. A qualitative summary of this result is as follows: Boundary variations on a scale l penetrate into the interior a distance of the order of l.

In the remainder of this notebook, we illustrate these ideas with graph sequences. In section 2, we consider a far-field approximation to the solution, and show that different boundary conditions can lead to the same far-field approximation. In section 3, we illustrate the limited penetration into the interior of high-frequency boundary components.

2. Far-Field Approximation

If we examine the field Φ some distance above the boundary $y = 0$, the higher harmonics will have essentially disappeared. Specifically, if we are at a height $y = L$, then the first harmonic has decayed by $e^{-\pi}$, the second harmonic by $e^{-2\pi}$, the third harmonic by $e^{-3\pi}$, etc. In the range $y \geq L$, it is reasonable to obtain a simple approximation to the solution by retaining only the first term in the series, and this is our far-field approximation, which we denote by Φ_f. It is given by

$$\Phi_f = C_1\, e^{-\pi y/L} \sin(\pi x/L)\ . \tag{4}$$

Of course if C_1 happens to be zero, then we would have to go to the next non-zero term of the series to get a useful approximation.

Far-field approximations are related to an interesting type of problem called an inverse problem. In general terms, an inverse problem is a situation in which we attempt to infer the sources of a field from measurements of the field outside the source region. In the present problem, the sources of the field are on the boundary $y = 0$ and $0 \le x \le L$ -- that is, we may think of the source in this problem as the specified function $f(x)$. Knowledge of f is tantamount to knowledge of C_n for all n. From our discussion leading to the far-field approximation (4), we see that as we move further from the boundary, we are less likely to be able to reconstruct f from our measurements. How well we do depends also on how accurately we can measure Φ.

It is clear from (4) that if we are sufficiently far from the boundary, two different sources f could look alike, if they have the same value of C_1. We construct an example illustrating this, in which we look simultaneously at three different solutions, corresponding to three different boundary conditions. In these calculations, we set $L = 1$. The three boundary condtions are f1[x] equal to 1, f2[x] equal to 2 on the left half interval and 0 on the right half interval, and f3[x] equal to $(4/\pi)\sin[\pi x/L]$.

```
L = 1.0;
```

```
f1[x_] := 1
```

```
f2[x_] := If[(x<=0.5),(2),(0)]
```

```
f3[x_] := (4/π)*Sin[π*x/L]
```

The Fourier sine coefficients for these functions are easily calculated. We have

```
C1[n_] := If[OddQ[n], 4/(n*π), 0]
```

```
C2[n_] := (4/(n*π)) * (1 - Cos[(n*π)/2])
```

```
C3[n_] := If[n == 1, 4/(n*π), 0]
```

These boundary functions have been chosen so that they have the same C_1:

```
C1[1]
```

$\frac{4}{\pi}$

```
C2[1]
```

$\frac{4}{\pi}$

```
C3[1]
```

$\frac{4}{\pi}$

Now we construct the solutions corresponding to each initial condition. For each solution, we include an argument k which is the number of terms to keep in the partial sum. It is worth noting that the exact solution Φ3 is also the far-field approximation for Φ1 and Φ2.

```
Φ1[x_,y_,k_] := Sum[C1[n]*Exp[-n*π*y/L]*
           Sin[n*π*x/L],{n,1,k}]
```

```
Φ2[x_,y_,k_] := Sum[C2[n]*Exp[-n*π*y/L]*
              Sin[n*π*x/L],{n,1,k}]

Φ3[x_,y_] := C3[1]*Exp[-π*y/L]*
              Sin[π*x/L]
```

Now we define the solution in a computationally convenient way. For $y = 0$, we use the given boundary functions. For $y < 0.1L$ we use 100 terms in the series, and for $y \geq 0.1L$ we use 10 terms in the series. A more sophisticated setup would actually calculate the number of terms needed for any given y.

```
sol1[x_,y_] := If[(y == 0),(f1[x]),
    (If[(y < 0.1*L),(Φ1[x,y,100]),(Φ1[x,y,10])])]

sol2[x_,y_] := If[(y == 0),(f2[x]),
    (If[(y < 0.1*L),(Φ2[x,y,100]),(Φ2[x,y,10])])]

sol3[x_,y_] := If[(y == 0),(f3[x]),(Φ3[x,y])]
```

Now we define a function field[y] which produces a graph of the three solutions at the height y. The first solution is plotted in red, the second in green and the third in blue.

```
field[y_] := Plot[{sol1[x,y],sol2[x,y],sol3[x,y]},
              {x,0,L},AxesLabel->{"x","Φ"},ImageSize->250,
              PlotLabel->Row[{"y = ",PaddedForm[y,{5,2}]}],
              PlotStyle->{{RGBColor[1,0,0],Thickness[0.004]},{RGBColor[0,1,0],Thickness[0.004]}
              RGBColor[0,0,1],Thickness[0.004]}},PlotRange->{0,2.01}]
```

We use a Do loop to construct a graph sequence in which y is increased gradually. We let y run from 0 to L in increments of $0.01*L$, thus 101 graphs. The printed version of this notebook shows only the initial graph in this sequence.

```
Do[Print[field[i*0.01]],{i,0,100}]
```



For visualization in the printed version of this notebook, we construct an abbreviated sequence of 6 graphs, with y running from 0 to 1 in increments of 0.2.

```
Do[Print[field[i*0.2]],{i,0,5}]
```